Finite Fourier Coefficients—FOUC

The program FOUC (**Fou**rier **C**oefficients) is a subroutine subprogram that calculates finite Fourier cos coefficients and finite Fourier sin coefficients when given N sample values x_m (m = 0, 1, 2,⋯, N −1) at equally spaced sample points. Unlike the Fast Fourier Transform, this program has a feature that there is no restriction on the number of sample values, and the number may be even or odd.

FOUC（Finite **Fou**rier **C**oefficients）

【Purpose】

To compute the finite Fourier coefficients for the given equally spaced data x_m (m =0, 1, 2,⋯, N-1).

【Usage】

（1） How to connect

```
CALL FOUC (N, X, ND1, A, B, ND2, NFOLD)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of data X	Unchanged
X	R 1-D array (ND1)	Equal interval data (ND1. GE. N)	Data normalized by maximum value
ND1	I	Dimension size of X in calling program	Unchanged
A	R 1-D array (ND2)	No need to input here	Representative value of relative amplitude
B	R 1-D array (ND2)	No need to input here	Probability distribution for each class (%)
ND2	I	Dimension size of A & B (ND .GE. N/2+1)	Average value normalized by maximum value
NFOLD	I	No need to input here	Number of coefficients

(2） Necessary subroutines and function subprograms

None

(3） Remarks

i) When you return from the subprogram, $B(1)$ will always be 0, and when N is even, $B(NFOLD)$ will also be 0.

ii) The argument *NFOLD* returns $N/2+1$ if N is even, or $(N+1)/2$ if N is odd.

【Calculation Method】

Given equally spaced data x_m ($m = 0, 1, 2, \cdots, N-1$), the Fourier cos coefficients A_k and Fourier sin coefficients B_k are calculated by the following equations.

i) N=Even number

$$A_k = \frac{2}{N}\sum_{m=0}^{N-1} x_m \cos\frac{2\pi km}{N} \qquad k = 0, 1, 2, \cdots, N/2-1, N/2$$

$$B_k = \frac{2}{N}\sum_{m=0}^{N-1} x_m \sin\frac{2\pi km}{N} \qquad k = 1, 2, \cdots, N/2-1$$

ii) N=Odd number

$$A_k = \frac{2}{N}\sum_{m=0}^{N-1} x_m \cos\frac{2\pi km}{N} \qquad k = 0, 1, 2, \cdots, (N-1)/2$$

$$B_k = \frac{2}{N}\sum_{m=0}^{N-1} x_m \sin\frac{2\pi km}{N} \qquad k = 1, 2, \cdots, (N-1)/2$$

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * *                    FOUC   1
C     SUBROUTINE FOR FOURIER COEFFICIENTS                             FOUC   2
C  * * * * * * * * * * * * * * * * * * * * * * *                    FOUC   3
C                                                                     FOUC   4
C                         CODED BY Y.OHSAKI                           FOUC   5
C                                                                     FOUC   6
C     PURPOSE                                                         FOUC   7
C       TO COMPUTE FINITE FOURIER SINE AND COSINE COEFFICIENTS FOR    FOUC   8
C       A SERIES OF EQUI-SPACED DATA                                  FOUC   9
C                                                                     FOUC  10
C     USAGE                                                           FOUC  11
C       CALL FOUC(N,X,ND1,A,B,ND2,NFOLD)                            FOUC  12
C                                                                     FOUC  13
C     DESCRIPTION OF ARGUMENTS                                        FOUC  14
C       N      - TOTAL NUMBER OF DATA                                 FOUC  15
C       X(ND1) - EQUI-SPACED DATA                                     FOUC  16
C       ND1    - DIMENSION OF X IN CALLING PROGRAM                    FOUC  17
C       A(ND2) - FOURIER COS COEFFICIENTS                             FOUC  18
C       B(ND2) - FOURIER SIN COEFFICIENTS                             FOUC  19
C       ND2    - DIMENSION OF A,B IN CALLING PROGRAM  ND2.GE.N/2+1    FOUC  20
C       NFOLD  - TOTAL NUMBER OF EACH COEFFICIENT                     FOUC  21
C                                                                     FOUC  22
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                   FOUC  23
C       NONE                                                          FOUC  24
C                                                                     FOUC  25
      SUBROUTINE FOUC(N,X,ND1,A,B,ND2,NFOLD)                          FOUC  26
C                                                                     FOUC  27
      DIMENSION X(ND1),A(ND2),B(ND2)                                  FOUC  28
      PARAMETER (P2=6.283185)                                         FOUC  29
C                                                                     FOUC  30
```

```
      NFOLD=N/2+1                                                    FOUC  31
      DO 120 K=1,NFOLD                                               FOUC  32
      AK=0.                                                          FOUC  33
      BK=0.                                                          FOUC  34
      DO 110 M=1,N                                                   FOUC  35
      AK=AK+X(M)*COS(P2*REAL((K-1)*(M-1))/REAL(N))                   FOUC  36
      BK=BK+X(M)*SIN(P2*REAL((K-1)*(M-1))/REAL(N))                   FOUC  37
  110 CONTINUE                                                       FOUC  38
      A(K)=2./REAL(N)*AK                                             FOUC  39
      B(K)=2./REAL(N)*BK                                             FOUC  40
  120 CONTINUE                                                       FOUC  41
      RETURN                                                         FOUC  42
      END                                                            FOUC  43
```

【Example】

Calculate the finite Fourier coefficients of the data given in the DATA statement.

```
C
      DIMENSION DATA(16),A(9),B(9)
      DATA  NN/16/,DATA/5.0,32.0,38.0,-33.0,
     &                  -19.0,-10.0,1.0,-8.0,-20.0,10.0,
     &                  -1.0,4.0,11.0,-1.0,-7.0,-2.0/
C
      CALL FOUC(NN,DATA,16,A,B,9,NFOLD)
      WRITE(6,601) NN,(K-1,A(K),B(K),K=1,NFOLD)
      STOP
  601 FORMAT('EXAMPLE WAVE'//'-- FINITE FOURIER COEFFICIENTS --'//T5,
     *       'TOTAL NUMBER OF DATA =',I3//T8,'K',TR10,'A',TR10,'B'//
     *      (I8,TR2,2F11.3) )
      END
```

Output :

```
EXAMPLE WAVE

-- FINITE FOURIER COEFFICIENTS --

    TOTAL NUMBER OF DATA = 16

       K           A           B

       0       0.000       0.000
       1       7.759      -4.143
       2       5.489       8.380
       3       4.958      11.952
       4      -6.750       8.750
       5      -4.188      -3.856
       6      -7.239      -2.370
       7       3.971      -4.951
       8       2.000       0.000
```